

TD 1 - Arithmétique : algorithme d'Euclide étendu

Soit a et b deux entiers naturels.

On note d leur PGCD. On cherche à déterminer un couple d'entiers (u, v) tels que : $au + bv = d$.

Algorithme d'Euclide sur un exemple :

	restes	u	v
	$a = 3575$	1	0
quotients	$b = 756$	0	1
4	551	1	-4
1	205	-1	5
2	141	3	-14
1	64	-4	19
2	13	11	-52
4	12	-48	227
1	$d = 1$	$u = 59$	$v = -279$

Travail demandé :

Écrire un script Scilab qui calcule les valeurs d , u et v à partir des entiers a et b .

1. Commencer par programmer une fonction qui prend en argument deux entiers naturels et qui renvoie le quotient de ces deux entiers.
2. On peut ensuite programmer l'algorithme d'Euclide en s'inspirant du tableau ci-dessus et en utilisant deux vecteurs lignes de dimensions 1×3 , initialisés par : $A = [a, 1, 0]$ et $B = [b, 0, 1]$. Calculer et ranger les lignes successives du tableau dans les variables vectorielles A et B . En sortie, l'affichage renvoie le vecteur : $[d, u, v]$.

Script Scilab

```
// Algorithme d'Euclide étendu
// -----
// fonction qui renvoie le quotient de deux entiers
function q=iquo(a, b)
    k=0;
    d=a;
    while d>=0
        k=k+1;
        d=d-b;
    end
    q=k-1;
endfunction
// -----
// algorithme d'Euclide étendu
a=input("entrer l'entier a : ");
b=input("entrer l'entier b : ");
// -----
A=[a,1,0];
B=[b,0,1];
q=iquo(A(1,1),B(1,1));
C=A-q.*B;
while C(1,1)<>0
    A=B;
    B=C;
    q=iquo(A(1,1),B(1,1));
    C=A-q.*B;
end
// affichage de d=PGCD(a,b), u et v
// tels que au+bv=d
disp(B);
```